

## Chapter 7: Acquiring Data

### Example 7-1

Use the code below to export the sales table. Then use the Import wizard to create the import specification called “Sales Import Specification” and import the file as SalesImport.

```
Public Sub ExportSales()
' This procedure exports the Sales table to a text file.
' You could do the same thing by using the TransferText
' action in a Microsoft Access macro.

    Dim strFile As String
    strFile = CurrentProject.Path & "\Sales.txt"

    ' Export the Sales table to Sales.txt in the same folder
    ' as this database. If you set up an export specification
    ' ahead of time, you can use it here. Since we did not,
    ' the text file will use Microsoft Access's default
    ' settings for acExportDelim to define the exported file's
    ' format.
    DoCmd.TransferText acExportDelim, , "Sales", strFile, True
End Sub
```

### Example 7-2

This code will ask the user for the text file to link. Then it will delete the existing SalesImport table in the database (if it exists, from Example 7-1) and replace it with the linked table SalesImport. Discuss the use of With...End With. Define what a dialog-type form is, how to create one, and why they are used.

```
Public Sub LinkFile()
' This procedure links a text file to the database. It
' uses the Application.FileDialog object to ask the user
' for the name of the file.
On Error GoTo Err_LinkFile

    Dim varFile

    ' Ask the user for the name of a text file to link.
    With Application.FileDialog(msoFileDialogOpen)

        ' Start the file with an S.
        .InitialFileName = "S*.txt"
        ' Clear filters and then add one for Text Files.
        ' Make it the first and only filter the user
        ' will see.
        .Filters.Clear
        .Filters.Add "Text Files", "*.txt", 1
```

## MICROSOFT ACCESS AND VBA: Building a Complete Solution

```
.FilterIndex = 1
' Add a title to the browser box and give the
' action button the caption Link.
.Title = "Select File to Link"
.ButtonName = "Link"
' If the user responded, then set the file
' variable to the first file in the
' Selected Items collection. If we had allowed
' more than one file selection, we would
' iterate through the collection to
' get a list of selected files.
If .Show Then
    varFile = .SelectedItems(1)
Else
    ' They did not select anything. Just
    ' exit the procedure.
    GoTo Exit_LinkFile
End If
End With

' Delete the existing linked table SalesImport.
DoCmd.DeleteObject acTable, "SalesImport"
' Link the file the user selected by using the
' Sales Import Specification (this must already
' have been created).
DoCmd.TransferText acLinkDelim, _
    "Sales Import Specification", "SalesImport", _
    varFile, True

Exit_LinkFile:
    On Error Resume Next
    Exit Sub
Err_LinkFile:
    If Err.Number = 7874 Then
        ' If the table we are trying to delete does
        ' not exist in the database, then just
        ' resume to the next line.
        Resume Next
    Else
        MsgBox Err.Number & " " & Err.Description, _
            vbCritical, "LinkFile"
        Resume Exit_LinkFile
    End If

End Sub
```